



MPC-3000のデータログとTCP Packetによる取得

- 温湿度計から1秒間隔で温度と湿度を読み、次のいずれかに任意の間隔で記録します。
 - (1) MPC-3000のポイントデータ (P())
 - (2) MMC (Multi Media Card) のファイル
 - (3) USBメモリのファイル
- それをPCのアプリケーションでダウンロード、保存します。
- アプリケーションが発行するコマンド、MPC側の処理はユーザーが作り込みます。
- アプリケーションは接続中、MPCの現在時刻、温湿度等をモニターしています。
- MMC、USBメモリのREAD/WRITEが衝突しないようにしなければなりません。
- サポート : MPC-3000 (SH2A) BL/I 2.01_39 2019/12/18 以降



■ アプリケーション実行例



ダウンロード (DL) したデータ (TextBox)

1データ (33byte) ずつDL

バルク (このサンプルでは14データ=462byte) でDL

データをクリップボードへコピー

データをPCにCSV保存

MPCのログデータを消去

接続状況、エラー等の情報表示

MPCの現在日時、現在温湿度、ログ数
これらは0.5秒間隔で読取

TextBox行数、データ数、最高最低温湿度、DL所要時間

(※1行DLは所要時間比較のため、TextBox行数はデバッグのため)

■ MPC記録時のフォーマット

- P() に記録する場合 : X()=年月日、Y()=時分秒、U()=温度湿度 をHEXで記録します。

例)

X(1) =&H20191223 Y(1) =&H00183050 U(1) =&H02510311
 2019年12月23日 18時30分50秒 251°C 311% (温度湿度は10倍の値)
 (10進表示では X(1)=538513955 Y(1)=1585232 U(1)=38863633)

これを文字列に変換して送信します。

```
FORMAT "0000-00-00"
TCP_X7$=HEX$(X(1))
FORMAT "00:00:00"
TCP_X7$=TCP_X7$+" "+HEX$(Y(1))
FORMAT "00.0"
TCP_X7$=TCP_X7$+" "+HEX$(U(1)>>16)+" "+HEX$(U(1)&&HFFFF)+"¥r¥n"
PR TCP_X7$
```

RUN

```
# TCP_X7$= 2019-12-23, 18:30:50, 25.1, 31.1 ←文字列
```

- MMC、USBメモリに記録する場合 : カンマ区切りの文字列としてファイルに追記します。

```
Date, Time, degC, %RH ←第1行目
2019-12-23, 16:29:04, 23.6, 39.6 ←年月日, 時分秒, 温度, 湿度
2019-12-23, 16:29:06, 23.6, 39.2
```

■ ダウンロード所要時間実測値(概数)

記録先	データ数	時間(Bulk読込) (ローカル接続, sec)	時間(1行ずつ読込) (ローカル接続, sec)	時間(Bulk読込) (外からInternet経由, sec)
P()	1000	2.0	15.8	
	5000	10.0	78.6	23.7
MMC	1000	1.6	15.7	
	5000	7.5	78.0	
USB	1000	50.9	62.9	
	5000	259.6	321.0	

- ・ CSVファイルの保存サイズ(Win10) 1000データ:30KB、5000データ:147KB。
- ・ MMCよりP()が遅いのは送信毎に数値→文字列変換を行うため。
- ・ 時間はパソコン、ネットワーク環境、ネットワーク状態等により変動します。
- ・ USBメモリはSeekタイムがボトルネックです。


■ ダウンロードサンプル

10秒間隔で3日半、約31500行 920KB のデータ。記録先はP()、ダウンロード時間66秒(ローカル接続)。

```
Date, Time, degC, %RH
2019-12-23, 18:30:50, 25.1, 31.1
2019-12-23, 18:31:00, 25.3, 30.6
2019-12-23, 18:31:10, 25.2, 30.6
2019-12-23, 18:31:20, 25.2, 30.6
2019-12-23, 18:31:30, 25.2, 30.6
2019-12-23, 18:31:40, 25.3, 30.5
2019-12-23, 18:31:50, 25.2, 30.9
2019-12-23, 18:32:00, 25.2, 30.7
2019-12-23, 18:32:10, 25.1, 30.8
2019-12-23, 18:32:20, 25.1, 30.6
2019-12-23, 18:32:30, 25.1, 30.9
2019-12-23, 18:32:40, 25.1, 31
2019-12-23, 18:32:50, 25.1, 30.8
2019-12-23, 18:33:00, 25.1, 30.9
2019-12-23, 18:33:10, 25.1, 30.9
2019-12-23, 18:33:20, 25, 30.8
2019-12-23, 18:33:30, 25, 30.9
2019-12-23, 18:33:40, 25, 30.8
2019-12-23, 18:33:50, 25, 30.7
2019-12-23, 18:34:00, 24.9, 30.9
2019-12-23, 18:34:10, 24.9, 31
2019-12-23, 18:34:20, 24.8, 31
```

(以下略)

表計算ソフトでグラフ化



■ MPCプログラム (MMC用)

• 設定、PACKET受信、コマンド解析 部分

```

/*****
/* MMC Log & Read
/*****
SET_IP 192 168 0 20 255 255 255 0 192 168 0 248
SET_IP TCP Telnet
SET_IP 115200 /* プログラムポート 115200bps
SET_IP TCP PACKET 50001 /* スレーブ設定
CNFG# 2 RS485 "9600b7pes1NONE" /* 温湿度計通信設定

DIM log_dat(15) /* 温湿度計からの受信データ
TCP_RO$=""
TCP_X7$="" /* 送信文字列のバッファとして使用 Max512byte
OFF 0 1 -1 /* 0:Running,1:Connecting,-1:Read/Write
Interlock

IF MMC(0) <> 1 THEN
  PR "MMC not mounted!!"
  END
END_IF

GOSUB *Log_ClearMemory
QUIT_FORK 1 *Log_Measure /* 温湿度計測タスク

DO
  IF TCP(0) == 1 THEN
    ON 1 /* 接続インジケータLED

    log_receive$ = TCP_RO$ /* Receive
    IF log_receive$ <> "" THEN

      TCP_RO$ = ""
      ptr_ = log_receive$
      log_command$ = PTR$(2) /* 受信文字列からコマンドを抽出
      log_parameter = VAL(log_receive$) /* パラメータ
      PR log_command$ log_parameter

      SELECT_CASE VOID
        CASE log_command$ == "RL" : GOSUB *Log_ReadLog
        CASE log_command$ == "CM" : GOSUB *Log_SetClearMemory
        CASE log_command$ == "CT" : GOSUB *Log_CheckTime
        CASE log_command$ == "CP" : GOSUB *Log_CheckDataCount
        CASE log_command$ == "HL" : GOSUB *Log_GetHighLow
        CASE log_command$ == "PO" : GOSUB *Log_StopMmcWrite
        CASE log_command$ == "FN" : GOSUB *Log_RefFileName
      CASE_ELSE
      END_SELECT

    END_IF
  ELSE
    OFF 1
  END_IF
  SWAP
LOOP

```

- ログデータを返すサブルーチン。
サンプルには3種類あり、コマンドのパラメータで選択されます。速いのは(2)か(3)です。

```

*Log_ReadLog                                /* ログデータを返す
IF log_parameter==0 THEN
  GOSUB *Log_ReadLogBlk1
  RETURN
END_IF
IF log_parameter>1 THEN
  GOSUB *Log_ReadLogBlk
  RETURN
END_IF

MMC_READ TCP_X7$                                /* (1) 1行ずつ返す
IF EOF(1)==0 THEN                                /* MMCから1行読込
  FORMAT ""
  TCP_X7$="EOF "+STR$(log_reccount)+" Datas"
  GOSUB *Log_TX0
  MMC_READ -1
  RETURN
END_IF
GOSUB *Log_TX0                                /* 1行送信
RETURN

*Log_ReadLogBlk                              /* (2) 指定行数ずつ返す
TCP_X7$=""
FOR i=1 TO log_parameter
  MMC_READ log_txbuf$
  IF EOF(1)==0 THEN
    TCP_X7$=TCP_X7$+"EOF "+STR$(log_reccount)+" Datas"
    MMC_READ -1
    BREAK                                     /* EXIT FOR
  END_IF
  TCP_X7$=TCP_X7$+log_txbuf$+"¥r¥n"
  SWAP
NEXT
GOSUB *Log_TX0
RETURN

*Log_ReadLogBlk1                            /* (3) 送信文字数を管理して返す
TCP_X7$=log_remain$                            /* 前回の残り
DO
  MMC_READ log_txbuf$
  log_txbuf$=log_txbuf$+"¥r¥n"
  IF EOF(1)==0 THEN
    TCP_X7$=TCP_X7$+"EOF "+STR$(log_reccount)+" Datas" /* ※
    MMC_READ -1
    log_remain$=""
    BREAK                                     /* EXIT DO
  END_IF
  IF IPC(TCP_X7$)+LEN(log_txbuf$)>480 THEN /* 上の ※ の文字長も考慮
    log_remain$=log_txbuf$
    BREAK                                     /* EXIT DO
  ELSE
    TCP_X7$=TCP_X7$+log_txbuf$
  END_IF
  SWAP
LOOP
PR "IPC(TCP_X7$)" IPC(TCP_X7$)
GOSUB *Log_TX0
RETURN

```

• その他のサブルーチン

```

*Log_SetClearMemory                                /* ログ消去フラグをセットする
  log_clearmemglf=1
  TCP_X7$="OK"
  GOSUB *Log_TX0
  RETURN

*Log_CheckTime                                     /* MPCの日時、温湿度と記録数を返す
  FORMAT ""
  TCP_X7$=DATE$(3)+" "+TIME$(1)
  TCP_X7$=TCP_X7$+" T"+log_tmp$+" H"+log_hum$
  TCP_X7$=TCP_X7$+" DataCount"+STR$(log_reccount)
  GOSUB *Log_TX0
  RETURN

*Log_CheckDataCount                               /* 記録数を返す
  FORMAT ""
  TCP_X7$=STR$(log_reccount)
  GOSUB *Log_TX0
  RETURN

*Log_GetHighLow                                   /* 最高最低温湿度を返す
  FORMAT "00.0"
  TCP_X7$="TH"+STR$(log_hitmp)+" , TL"+STR$(log_lotmp)
  TCP_X7$=TCP_X7$+"/HH"+STR$(log_hihum)+" , HL"+STR$(log_lohum)
  GOSUB *Log_TX0
  RETURN

*Log_StopMmcWrite                                 /* MMC読取中は書込を抑制
  IF log_parameter>0 THEN
    ON -1
  ELSE
    OFF -1
  END IF
  TIME 1000
  TCP_X7$="OK"
  GOSUB *Log_TX0
  RETURN
/* Consider to MMC Seek Time

*Log_RefFileName                                  /* MMC保存ファイル名を返す
  TCP_X7$=FILE$
  GOSUB *Log_TX0
  RETURN

*Log_TX0                                           /* TCP Packet送信
  TCP_R0$=""
  TCP_X0$=TCP_X7$+"¥r¥n"
  WAIT TCP_X0$=""
  TCP_X7$=""
  RETURN
/* ※ LEN("¥r¥n")==4

```

• 温湿度を計測してMMCのファイルに追記するタスク

```

*Log_Measure /* 温湿度計測タスク

DO
  INPUT# 2 CLR_BUF /* 受信バッファクリア
  PRINT# 2 CHR$(5) "02" CHR$(2) "RPV01" CHR$(3) "¥r¥n" /* 温湿度計にコマンド送信
  INPUT# 2 EOL|10 log_input$ /* 温湿度計データ受信
  /* 受信データ例 {06}01{02}APV01=2001, 1,01, 0,50,0,0,0, 25.3,0, 33.8{03}
  log_dat(0)=VAL(log_input$)
  FOR i_1 TO 14 /* log_dat()に数値を入れていく
    log_dat(i_1)=VAL(0)
  NEXT

  FORMAT "00.0"
  log_tmp=log_dat(10)*10+log_dat(11) /* 温度
  IF log_tmp>log_hitmp THEN /* 最高温度
    log_hitmp=log_tmp
  END_IF
  IF log_tmp<log_lotmp THEN /* 最低温度
    log_lotmp=log_tmp
  END_IF
  log_tmp$=STR$(log_tmp)
  log_hum=log_dat(13)*10+log_dat(14) /* 湿度
  IF log_hum>log_hihum THEN /* 最高湿度
    log_hihum=log_hum
  END_IF
  IF log_hum<log_lohum THEN /*最低湿度
    log_lohum=log_hum
  END_IF
  log_hum$=STR$(log_hum)

  log_tim$=TIME$(1)
  WHILE log_tim$==TIME$(1) /* 1sec経過待ち
    log_nowtime=TIME(0) /* HEX
    IF log_clearmemlf==1 THEN
      GOSUB *Log_ClearMemory
    END_IF

    log_interval=&H01 /* 2seconds
    /* log_interval=&H0F /* 10seconds
    /* log_interval=&HFF /* 1minute
    IF log_nowtime&log_interval==0 THEN
      FORMAT ""
      log_log$=DATE$(3)
      log_log$=log_log$+", "+TIME$(1)
      FORMAT "00.0"
      log_buf1$=STR$(log_tmp)
      log_buf2$=STR$(log_hum)
      log_log$=log_log$+", "+log_buf1$+", "+log_buf2$

      IF SW(-1)==0 THEN /* TCP通信タスクとのインターロック
        MMC_WRITE log_log$
        SWAP /* (Telnet)PRINT中のタスク遷移回避のため
        PRINT "MMC_WRITE" log_log$
        INC log_reccount
      END_IF


      WAIT (TIME(0)&log_interval)<>0
    END_IF
    SWAP
  WEND
  GOSUB *Log_Led_Flick0
LOOP

*Log_ClearMemory
log_reccount=0
log_hitmp=0
log_lotmp=1000
log_hihum=0
log_lohum=1000
log_clearmemlf=0
FILE$=DATE$(0)+".CSV"
REMOVE FILE$
MMC_WRITE "Date,Time,degC,%RH"
MMC_READ -1
log_remain$=""
RETURN


*Log_Led_Flick0
IF IN(0)&&H1<>0 THEN
  OFF 0
ELSE
  ON 0
END_IF
RETURN

```

■ Androidタブレットの簡単なTCPアプリを作ってみました。
MPCの現在の年月日時刻、温度湿度、Logしたデータ数を表示します。



Microsoft Visual Studio 2019 Community、Xamarin Form で作成



← MPCのDate Time 現在温度湿度 Logデータ数
← 最高最低 温度/湿度

MPCにコマンドを送信して返信された文字列をそのまま表示しています。